Buoyant Flexure in Glacier Calving

Reference: “On the role of buoyant flexure in glacier calving”, Wagner et al (2016), Geophys. Res. Lett., 43, GL0670247

In this notebook we solve the dimensionless form of the floating beam equation -- Eq (2) in the paper, subject to boundary conditions (3).
The shear force condition at $X = X_t$, can be set to $W'''(X_t) = 0$, as in (3). Alternatively, a nonzero force Q, corresponding to an unbalanced protrusion can be added (see section 2.5).

The method here is to find a general (analytic) solution for W(X), using 4 of the 6 conditions, for unkown x_g and τ.
Then we determine x_g and τ using the remaining two conditions. Since x_g and τ depend on each other, we first solve for $x_g(\tau)$ and $\tau(x_g)$ independently and then compute $x_g(X_t, Q)$.

In[135]:=
```
Clear[a, b, S, d, Xg, Xt, Q, XgAn, plotW]
```

```
W[X_, τ_, Xg_, Xt_, Q_] =
  w[X] /. DSolve[{w''''[X] + τ w''[X] + w[X] == 0, w[Xg] == d - S Xg,
     w[Xt] == -(b + τ) / d, w''[Xg] == 0, w'''[Xt] == Q}, w[X], X][[1]];
DW[X_, τ_, Xg_, Xt_, Q_] = D[W[X, τ, Xg, Xt, Q], X];
DDW[X_, τ_, Xg_, Xt_, Q_] = D[D[W[X, τ, Xg, Xt, Q], X], X];
Xg0[Xt_, τ_, Q_] :=
  Xg /. FindRoot[Re[DW[Xg, τ, Xg, Xt, Q]] + S == 0, {Xg, XgAn + 1}][[1]];
τ0[Xt_, Xg_, Q_] :=
  τ /. FindRoot[Re[DDW[Xt, τ, Xg, Xt, Q]] == a + b W[Xt, τ, Xg, Xt, Q], {τ, 0}][[1]];
Xg[Xt_, Q_] := Module[{}, err = 1; err1 = err; dXg = 0.5; XgZ = XgAn;
  While[Abs[err] > .001, If[err * err1 < 0, dXg = -dXg / 2]; XgZ = XgZ + dXg;
   τZ = τ0[Xt, XgZ, Q]; err1 = err; err = Xg0[Xt, τZ, Q] - XgZ]; XgZ]

g = 9.81 ; (*gravity*) v = 0.3 ; (*Poisson Ratio*)
rw = 1000; (*kg/m^3, density of water*)
h = 100; (*beam thickness*) S = 0.1; (*bed slope*)
```

$$Y = 1\times 10^7;\ (*\text{Young's Modulus}*)\ B = \frac{Y h^3}{12\left(1 - v^2\right)};$$

$$(*\text{Bending Stiffness}*)\ lw = \left(\frac{B}{rw\, g}\right)^{\frac{1}{4}};\ (*\text{buoyancy length}*)$$

$$H = h / lw;\ d = 0.9 H;\ (*\rho_i/\rho_w\ h*)$$

$$a = \frac{1}{12} d\left(H^2 - 3 d H + 2 d^2\right);\ b = \frac{1}{2} d\,(H - d);$$

(*for derivation of a and b, see below*)

$(*a = -0.006\ H^3; b = 0.045\ H^2*)$ (* assuming d = 0.9H *)

$XgAn = d / S - \sqrt{2};$ (*analytic location of x_g in the limit $X_t \to \infty$ *)

```
Winf = H / 2 - d ;   (*distance between sea level and isostatic draft height*)
```

Xt = XgAn + 3; (*freely chosen fixed position X_t*)

```
Q = -0.0425;  (*dimensionless shear force; -ve for upward force;
set to zero in case of no protrusion at terminus*)
```

Xg1 = Xg[Xt, Q]; (*compute x_g for X_t and Q*)

τ1 = Re[τ0[Xt, Xg1, Q]]; (*compute τ for X_t and Q*)

Out[143]= 7.65019

Out[144]= -0.0546199

```
plotW[Xt_, Q_, Hu_] := Module[{}, Xg1 = Xg[Xt, Q]; τ1 = Re[τ0[Xt, Xg1, Q]] ;
  p = Plot[Re[W[X, τ1, Xg1, Xt, Q]], {X, Xg1, Xt}, PlotStyle → Hue[Hu]]; p]
bedplot[Xt_] := Plot[d - S X, {X, Xg[Xt, 0] - 1, Xg[Xt, 0] + 3},
  PlotStyle → Black, PlotRange → All, Frame → True, FrameLabel → {"X", "W"}]
```

Plot the deflected profile for fixed X_t and 5 different values of Q ranging from -0.2 (cyan) to 0 (orange). The bed is in black.

```
Show[bedplot[Xt], Table[plotW[Xt, -(Qi - 1) / 20, Qi / 10], {Qi, 1, 5}], Axes -> False]
```

Out[151]=



Derivation of constants *a* and *b* (following Reeh 1968)



In the following w = w_t (i.e. the deflection at the glacier terminus)

p(ice) = $\rho_i\, g\, (z - h/2) = \rho_w\, g\, (d/h)\, (z - h/2)$
p(water) = $\rho_w\, g\, (d - h/2 - w - z)\, \Theta\, (d - h/2 - w - z)$

The in-plane stress is then given by

$$T = \int_{-h/2}^{h/2} p\, dz$$

and the moment at x_t is

$$M = \int_{-h/2}^{h/2} p\, z\, dz$$

```
σ = Simplify[ρw g d / h Integrate[z - h / 2, {z, -h / 2, h / 2}] +
   ρw g Integrate[d - h / 2 - w - z, {z, -h / 2, d - h / 2 - w}]]
```

$$\frac{1}{2} g \left(d^2 + w^2 - d\, (h + 2 w)\right) \rho_w$$

$T = \rho_w\, g\, \frac{1}{2} \left(d^2 - d\, h - 2\, d\, w\right)$ [ignoring the w^2 term]

```
M = Simplify[ρw g d / h Integrate[(z - h / 2) z, {z, -h / 2, h / 2}] +
   ρw g Integrate[(d - h / 2 - w - z) z, {z, -h / 2, d - h / 2 - w}]]
```

$$\frac{1}{12} g \left(2\, d^3 - 3\, d^2\, (h + 2\, w) - w^2\, (3\, h + 2\, w) + d\, \left(h^2 + 6\, h\, w + 6\, w^2\right)\right) \rho_w$$

$M = \frac{1}{12} \rho_w\, g \left(2\, d^3 - 3\, d^2\, h - 6\, d^2\, w + d\, h^2 + 6\, d\, h\, w\right) = \frac{1}{12} \rho_w\, g\, d \left(2\, d^2 - 3\, d\, h + h^2 + 6\, w\, (h - d)\right)$ [ignoring higher order w terms]

Simplify and non-dimensionalize:

$\tau = T \left(l_w{}^2 / B\right) = \frac{1}{2} D\, (D - H - 2\, W) = -b - D\, W$, where $b = \frac{1}{2} D\, (H - D) = 0.045\, H^2$

$\rightarrow W = -\frac{b+\tau}{D}$

$M = B\, w'' = \frac{B}{l_w} W'' = \frac{1}{12} \rho_w\, g\, l_w{}^3\, D \left(2\, D^2 - 3\, D\, H + H^2 + 6\, W\, (H - D)\right)$

$W''(X_t) = \frac{1}{12} D \left(2\, D^2 - 3\, D\, H + H^2\right) + \frac{1}{2} D\, W\, (H - D)) = \frac{1}{12} D \left(H^2 - 3\, D\, H + 2\, D^2\right) + b\, W = a + b\, W$,
where $a = \frac{1}{12} D \left(H^2 - 3\, D\, H + 2\, D^2\right) = -0.006\, H^3$